
How to Limit User Input in VB Combo Box with SendMessage API

Article ID: Q72677 

The information in this article applies to: 

- Standard and Professional Editions of Microsoft Visual Basic for   Windows, versions 2.0 and 3.0

- Microsoft Visual Basic programming system for Windows, version 1.0 

SUMMARY

You can specify a limit to the amount of text that can be entered into a combo box by calling SendMessage (a Windows API function) with the EM_LIMITTEXT constant. 

MORE INFORMATION

The following method can be used to limit the length of a string entered into a combo box. Check the length of a string inside a KeyPress event for the control, if the length is over a specified amount, then the formal argument parameter KeyAscii will be set to zero. 

Or, the preferred method of performing this type of functionality is to use the SendMessage API function call. After you set the focus to the desired edit control, you must send a message to the window's message queue that will reset the text limit for the control. The argument EM_LIMITTEXT, as the second parameter to SendMessage, will set the desired text limit based on the value specified by the third arguments. The SendMessage function requires the following parameters for setting the text limit: 

   SendMessage (hWnd%,EM_LIMITTEXT, wParam%, lParam)

   wParam%   Specifies the maximum number of bytes that can be

             entered. If the user attempts to enter more characters,

             the edit control beeps and does not accept the characters.

             If the wParam parameter is zero, no limit is imposed on

             the size of the text (until no more memory is available).

   lParam    Is not used.

The following steps can be used to implement this method: 

1. Create a form called Form1. 

2. Add a combo box called Combo1 to Form1. 

3. Add the following code to the general declarations section of Form1: 

    '*** Note: Each Declare statement must be on just one line:

    Declare Function GetFocus% Lib "user" ()

    Declare Function SendMessage& Lib "user" (ByVal hWnd%,

                                              ByVal wMsg%,

                                              ByVal wParam%,

                                              lp As Any)

    Const WM_USER = &H400

    Const EM_LIMITTEXT = WM_USER + 21

4. Add the following code to the Form_Load event procedure: 

    Sub Form_Load ()

       Form1.Show            ' Must show form to work on it.

       Combo1.SetFocus       ' Set the focus to the list box.

       cbhWnd% = GetFocus()  ' Get the handle to the list box.

       TextLimit% = 5        ' Specify the largest string.

       retVal& = SendMessage(cbhWnd%, EM_LIMITTEXT, TextLimit%, 0)

    End Sub

5. Run the program and enter some text into the combo box. You will 

   notice that you will only be able to enter a string of five

   characters into the combo box.

How to Programmatically Display or Hide a VB Combo Box List

Article ID: Q85991 

The information in this article applies to: 

· Standard and Professional Editions of Microsoft Visual Basic for Windows, versions 2.0 and 3.0 

· Microsoft Visual Basic programming system for Windows, version 1.0 

SUMMARY

The list of a Visual Basic drop-down combo box (Style=0 or 2) is usually opened and closed by using a mouse. However, you can also open and close the list of a combo box programmatically by using the Windows SendMessage function as described below. 

However, there is an easy way. In visual basic, when a drop-down combo box has the focus, you can press ALT-DOWN to open it up. Therefore, you can use SendKeys to send these keys to the combo box, as in this example: 

To do this programmatically, use the following code to change focus to the combo box and send the necessary keystroke: 

   combo1.SetFocus

   SendKeys "%{Down}"

MORE INFORMATION

The CB_SHOWDROPDOWN constant can be used with the SendMessage function to programmatically open or close the list of a Visual Basic drop-down combo box of Style=0 or Style=2 (Style=1 always has the list displayed). The following steps demonstrate how to open the list of a drop-down combo box: 

1. Start a new project in Visual Basic. Form1 is created by default. 

2. Place a combo box (Combo1) and a command button (Command1) on Form1. 

3. Add the following declarations and constants to the general 

   Declarations section of Form1:

   ' Enter each Declare statement as one, single line:

   Declare Function GetFocus Lib "User" () as Integer

   Declare Function GetParent Lib "User" (ByVal hWnd as Integer)

      as Integer

   Declare Function SendMessage Lib "User" (ByVal hWnd as Integer,

      ByVal wMsg as Integer, ByVal wParam as Integer,

      ByVal lParam as Any) as Long

   Global Const WM_USER = &H400

   Global Const CB_SHOWDROPDOWN = WM_USER + 15

4. Add the following code to the Form1 Load event procedure to put 

   some items in the combo box:

   Sub Form_Load ()

      Combo1.AddItem "apple"

      Combo1.AddItem "orange"

      Combo1.AddItem "banana"

   End Sub

5. Add the following code to the Command1_Click event procedure: 

   Sub Command1_Click ()

      Combo1.SetFocus

      cbhWnd% = GetFocus ()

      cblisthWnd% = GetParent (cbhWnd%)

      cbFunc% = -1           'cbFunc% = -1 displays the list

                             'cbFunc% = 0 hide the list

      retval& = SendMessage (cblisthWnd%, CB_SHOWDROPDOWN, cbFunc%, 0&)

   End Sub

6. Press the F5 key to run the program. Click Command1 to display the list 

   of the combo box.

If Style=2 for the combo box, there is no need to use the GetParent function. Use the return value of the GetFocus (cbhWnd% in the above example) call as the first parameter of the SendMessage function. 

NOTE: The list of a combo box with Style=0 or 2 will close when the combo box loses focus. 

How to Quickly List the Contents of a Directory

Article ID: Q119396

The information in this article applies to: 

· Standard and Professional Editions of Microsoft Visual Basic for Windows, version 3.0 

SUMMARY

Although Visual Basic has both built-in functions (Dir$ and Dir) and a built-in control (File list box) for listing the files in a directory, using a standard list box and the Windows API SendMessage function lists files in a directory much more quickly. 

NOTE: Replacing the File list box or Dir command using the technique described in this article will only improve performance perceptibly if the File list box or Dir command is used heavily in the application. 

MORE INFORMATION

This technique uses the LB_DIR message for list boxes to add a list of filenames to a list box. To do this, call the Windows API SendMessage function. The SendMessage function requires the following parameters to list the files: 

   SendMessage (hWnd%, LB_DIR, wParam%, lParam)

where 

   hWnd%    is the handle of the list box.

   wParam%  is an integer that specifies the permissible file

            attributes.

   lParam   is a long pointer to a string containing the full path and

            file name to add to the list. This can include wildcards.

The wParam% argument can be any combination of the following values: 

Value             Meaning

DDL_ARCHIVE       File has been archived.

DDL_DIRECTORY     File must be a directory name.

DDL_EXCLUSIVE     Exclusive flag. If the exclusive flag is set, only

                  files of the specified type are listed. Otherwise,

                  files of the specified type are listed in addition to

                  files that do not match the specified type.

DDL_HIDDEN        File must be hidden.

DDL_READONLY      File must be read only.

DDL_READWRITE     File can be read from or written to.

DDL_SYSTEM        File must be a system file.

Step-by-Step Example

1. Start Visual Basic; or, choose New Project from the File menu 

   (ALT+F, N) if Visual Basic is already running. Form1 is created by

   default.

2. Add a list box (List1) to Form1. 

3. Add a new code module to the project by choosing New Module from the 

   File menu.

4. Add the following code to the new code module: 

   ' Enter the Declare statement on one single line:

   Declare Function SendMessage Lib "user" (ByVal hwnd As Integer,

                                            ByVal wMsg As Integer,

                                            ByVal wParam As Integer,

                                            lParam As Any) As Long

5. Declare the following constants in the General Declaration section of 

   the code module

      Global Const WM_USER = &H400

      Global Const LB_DIR = WM_USER + 14

      Global Const DDL_ARCHIVE = &H0020

      Global Const DDL_DIRECTORY = &H0010

      Global Const DDL_EXCLUSIVE = &H8000

      Global Const DDL_HIDDEN = &H0002

      Global Const DDL_READONLY = &H0001

      Global Const DDL_READWRITE = &H0000

      Global Const DDL_SYSTEM = &H0004

6. To fill the list box with all of the files, which may or may not be 

   hidden, in the C:\WINDOWS directory, place the following code in the

   Form_Load event of Form1:

      Sub Form_Load ()

         List1.Clear

         s$ = "C:\WINDOWS\*.*"

         i% = SendMessage(List1.hWnd, LB_DIR, DDL_HIDDEN, ByVal s$)

      End Sub

7. Press the F5 key to run the application. 

The contents of the list box can be manipulated by using any of the standard list-box methods. In cases where the filenames are being used elsewhere in the code (for example, in an outline control), the Visible property of the list box can be set to False so that the list box will not be displayed. 

How to Quickly Search a List Box

Article ID: Q141027

4.00 WINDOWS 

  kbhowto

The information in this article applies to: 

· Standard, Professional, and Enterprise Editions of Microsoft Visual Basic, 16-bit and 32-bit, for Windows, version 4.0 

SUMMARY

A popular item in a user interface is to "link" a text box to a list box so that as the user types text into the text box, the nearest match in the list box is selected. Although this technique can be implemented fairly easily using pure Visual Basic code, the Windows API provides a quick and easy way of doing this. 

MORE INFORMATION

The following technique uses the LB_FINDSTRING message for a list box to locate a partial match for a string in a list box. To do this, call the Windows API SendMessage function. The SendMessage function requires the following parameters to list the files: 

SendMessage   (hWnd,LB_FINDSTRING, wParam, lparam)

where 

hWnd    is the hWnd of the list box.

wParam is an integer that specifies the starting point for the search. 

          Use -1 to search the whole list box.

Lparam   is a long pointer to the string to find.

Example to Demonstrate Searching a List Box

1. Start a new project in Visual Basic. Add a text box (Text1) and a list 

   box (list1) to Form1.

2. On the Visual Basic Insert menu, choose module to add a new code module 

   to the project. Add the following code to the General declarations

   section of module1:

      #If Win32 Then

         Declare Function SendMessage Lib "user32" Alias "SendMessageA" _

         (ByVal hwnd As Long, ByVal wMsg As Long, ByVal wParam As _

         Integer, ByVal lParam As Any) As Long

      #Else

         Declare Function SendMessage Lib "user" (ByVal hwnd As Integer, _

         ByVal wMsg As Integer, ByVal wp As Integer, lp As Any) As Long

      #End If

      Public Const LB_FINDSTRING = &H18F

3. In the Form_Load method for the form, add the following items to the 

   list box using the following code:

      Sub Form_Load()

         List1.Clear

         List1.AddItem "Apples"

         List1.AddItem "Banana"

         List1.AddItem "Bread"

         List1.AddItem "Break"

      End Sub

4. In the Change method of the text box, add the following code: 

      Sub Text1_Change()

         List1.ListIndex = SendMessage(List1.hWnd, LB_FINDSTRING, -1, _

         ByVal Cstr(Text1.Text))

      End Sub

If you run the code, typing text into the text box selects the first item in the list that matches the text in the text box. 

How to Read a Large File into Memory by Calling API Functions

Article ID: Q100513

The information in this article applies to: 

- The Visual Basic programming system for Windows, versions 2.0 and 3.0 - Microsoft Windows, version 3.1 or higher 

SUMMARY

This article demonstrates how to call Windows API functions to read a file of any size (including a huge file such as a bitmap) into memory and how to write a block of memory (including a huge memory block) out to a file. 

The information in this article applies only to Windows version 3.1 or higher because it uses Windows API functions introduced in Windows version 3.1. 

MORE INFORMATION

Perform the following steps to create a sample program that demonstrates how to read a large file into memory and write that memory back out to a file: 

1. Start Visual Basic or from the File menu, choose New Project (ALT, F, N) 

   if Visual Basic is already running. Form1 is created by default.

2. From the File menu, choose New Module (ALT, F, M). Module1 is created. 

3. Add the following code to the general-declarations section of Module1: 

   ' OpenFile() Structure

   Type OFSTRUCT

      cBytes As String * 1

      fFixedDisk As String * 1

      nErrCode As Integer

      reserved As String * 4

      szPathName As String * 128

   End Type

   ' OpenFile() Flags

   Global Const OF_READ = &H0

   Global Const OF_WRITE = &H1

   Global Const OF_READWRITE = &H2

   Global Const OF_SHARE_COMPAT = &H0

   Global Const OF_SHARE_EXCLUSIVE = &H10

   Global Const OF_SHARE_DENY_WRITE = &H20

   Global Const OF_SHARE_DENY_READ = &H30

   Global Const OF_SHARE_DENY_NONE = &H40

   Global Const OF_PARSE = &H100

   Global Const OF_DELETE = &H200

   Global Const OF_VERIFY = &H400

   Global Const OF_CANCEL = &H800

   Global Const OF_CREATE = &H1000

   Global Const OF_PROMPT = &H2000

   Global Const OF_EXIST = &H4000

   Global Const OF_REOPEN = &H8000

   ' Enter each of the following Declare statements on one, single line:

   Declare Function OpenFile Lib "Kernel" (ByVal lpFilename As

      String, lpReOpenBuff As OFSTRUCT, ByVal wStyle As Integer) As Integer

   Declare Function hRead Lib "kernel" Alias "_hread" (ByVal hFile As

      Integer, lpMem As Any, ByVal lSize As Long) As Long

   Declare Function hWrite Lib "Kernel" Alias "_hwrite" (ByVal hFile

      As Integer, lpMem As Any, ByVal lSize As Long) As Long

   Declare Function lClose Lib "kernel" Alias "_lclose" (ByVal hFile

      As Integer) As Integer

   ' Global Memory Flags

   Global Const GMEM_FIXED = &H0

   Global Const GMEM_MOVEABLE = &H2

   Global Const GMEM_NOCOMPACT = &H10

   Global Const GMEM_NODISCARD = &H20

   Global Const GMEM_ZEROINIT = &H40

   Global Const GMEM_MODIFY = &H80

   Global Const GMEM_DISCARDABLE = &H100

   Global Const GMEM_NOT_BANKED = &H1000

   Global Const GMEM_SHARE = &H2000

   Global Const GMEM_DDESHARE = &H2000

   Global Const GMEM_NOTIFY = &H4000

   Global Const GMEM_LOWER = GMEM_NOT_BANKED

   Global Const GHND = (GMEM_MOVEABLE Or GMEM_ZEROINIT)

   Global Const GPTR = (GMEM_FIXED Or GMEM_ZEROINIT)

   ' Enter each of the following Declare statements on one, single line:

   Declare Function GlobalAlloc Lib "Kernel" (ByVal wFlags As

       Integer, ByVal dwBytes As Long) As Integer

   Declare Function GlobalLock Lib "Kernel" (ByVal hMem As Integer)

       As Long

   Declare Function GlobalUnlock Lib "Kernel" (ByVal hMem As Integer)

       As Integer

   Declare Function GlobalFree Lib "Kernel" (ByVal hMem As Integer)

       As Integer

4. Add the following code to the Form_Load event procedure of Form1: 

   Sub Form_Load ()

      Dim InpFile As String

      Dim OutFile As String

      Dim hFile As Integer

      Dim fileStruct As OFSTRUCT

      Dim FSize As Long

      Dim BytesRead As Long

      Dim BytesWritten As Long

      Dim hMem As Integer

      Dim lpMem As Long

      Dim r As Integer

      Me.Show

      'Insert the name of a bitmap or file that is greater than 64K.

      '256COLOR.BMP is less than 5K in size, however, the routine

      'below still demonstrates how to read and write a file of any

      'size

      InpFile = "C:\WINDOWS\256COLOR.BMP"

      OutFile = "C:\WINDOWS\TEST.BMP"

      'Get the size of the file to be read

      FSize = FileLen(InpFile)

      If FSize > 0 Then

         'Allocate a block of memory equal to the size of the input file.

         hMem = GlobalAlloc(GPTR, FSize)

         If hMem <> 0 Then

            lpMem = GlobalLock(hMem)

            'Read the file into memory

            hFile = OpenFile(InpFile, fileStruct, OF_READ Or

                             OF_SHARE_DENY_NONE)

            BytesRead = hRead(hFile, ByVal lpMem, FSize)

            MsgBox Format(BytesRead) & " bytes read into memory"

            r = lClose(hFile)

            'Write the file back to disk to verify the file was

            'read correctly

            hFile = OpenFile(OutFile, fileStruct, OF_CREATE Or

                             OF_WRITE Or OF_SHARE_DENY_NONE)

            BytesWritten = hWrite(hFile, ByVal lpMem, FSize)

            MsgBox Format(BytesWritten) & " bytes written to output file"

            r = lClose(hFile)

            'Free resources

            r = GlobalUnlock(hMem)

            r = GlobalFree(hMem)

         Else

             MsgBox "Not enough memory to store file"

         End If

      Else

         MsgBox "Input file is zero bytes in length"

      End If

      End

   End Sub

5. From the Run menu, choose Start (ALT, R, S) or press F5 to run the 

   program. Form1 will be displayed and the program will end.

6. Use PaintBrush or some other bitmap editor to open C:\WINDOWS\TEST.BMP 

   to verify that it is the same bitmap as C:\WINDOWS\256COLOR.BMP.

